

比率のP値函数達

- 黒木玄
- 2022-09-08
- nbviewer版 (<https://nbviewer.org/github/genkuroki/public/blob/main/0036/P-value%20functions%20of%20proportions.ipynb>)
- repository版 (<https://github.com/genkuroki/public/blob/main/0036/P-value%20functions%20of%20proportions.ipynb>)
- ダウンロードPDF (<https://github.com/genkuroki/public/raw/main/0036/P-value%20functions%20of%20proportions.pdf>)
- PDF版(repository) (<https://github.com/genkuroki/public/blob/main/0036/P-value%20functions%20of%20proportions.pdf>)

このノートでは A の確率を $P(A)$ と表し, 確率変数 X の函数 $f(X)$ の期待値を $E[f(X)]$ と表す.

予備知識: 二項分布, 正規分布, ベータ分布. 終わりの方ではベイズ統計.

目次

- [1 データと統計モデルの設定](#)
- [2 P値函数と信頼区間の一般論](#)
- ▼ [3 累積分布函数 cdf と分位点函数 quantile](#)
 - [3.1 補累積分布函数 ccdf と補分位点函数 cquantile](#)
- ▼ [4 WilsonのP値と信頼区間](#)
 - [4.1 WilsonのP値](#)
 - [4.2 Wilsonの信頼区間](#)
 - [4.3 WilsonのP値と信頼区間の実装例](#)
- ▼ [5 WaldのP値と信頼区間](#)
 - [5.1 WaldのP値](#)
 - [5.2 Waldの信頼区間](#)
 - [5.3 WaldのP値と信頼区間の実装例](#)
- ▼ [6 Clopper-PearsonのP値と信頼区間](#)
 - [6.1 Clopper-PearsonのP値](#)
 - [6.2 二項分布とベータ分布の関係](#)
 - [6.3 Clopper-Pearsonの信頼区間](#)
 - [6.4 Clopper-PearsonのP値と信頼区間の実装例](#)
- ▼ [7 SterneのP値と信頼区間](#)
 - [7.1 SterneのP値と信頼区間](#)
 - [7.2 SterneのP値と信頼区間の実装例](#)
- [8 4種のP値函数と信頼区間を同時にプロット](#)
- ▼ [9 ベイズ統計での信用区間に対応するP値函数](#)
 - [9.1 二項分布モデルのHDI版信用区間](#)
 - [9.2 HDI版信用区間に対応するP値函数](#)
 - [9.3 HDI版の信用区間とベイズ版P値函数の実装例](#)
 - [9.4 WilsonのP値函数とベイズ版P値函数のHDI版の比較](#)
 - [9.5 ETI版の信用区間とそれに対応するP値函数](#)
 - [9.6 ETI版の信用区間とそれに対応するベイズ版P値函数の実装例](#)
 - [9.7 WilsonのP値函数とベイズ版P値函数のETI版の比較](#)
 - [9.8 道具を主義の色眼鏡で見ることの不合理性](#)

```
In [1]: 1 using Distributions
        2 using StatsPlots
        3 default(fmt=:png, size=(500, 300),
        4         titlefontsize=10, tickfontsize=6, guidefontsize=9)
        5
        6 # Sterneの信頼区間の定義で使用
        7 using Roots
        8 using StatsFuns: logit
        9
       10 # highest density interval (HDI) の構成で使用
       11 using Optim
```

1 データと統計モデルの設定

当たりまたは外れの目が出るルーレットを n 回まわしたとき当たりが k 回出たと仮定する.

このとき「 n 回中当たりが k 回」をこのルーレットに関するデータと呼ぶ.

そのデータの確率的揺らぎのモデルとして、二項分布を考える。

すなわち、次の確率質量関数で定義される k に関する確率分布を考える：

$$P(k|n, p) = \binom{n}{k} p^k (1-p)^{n-k} \quad (k = 0, 1, \dots, n).$$

ここで、 $0 \leq p \leq 1$ であり、 $\binom{n}{k}$ は次で定義される二項係数である：

$$\binom{n}{k} = \frac{n(n-1) \cdots (n-k+1)}{k!} = \frac{n!}{k!(n-k)!}.$$

2 P値関数と信頼区間の一般論

以下では具体的な数値 p_0 に関する仮説「当たりが出る確率は p である」と「 n 回中当たりが k 回」というデータの **両立性** (compatibility) もしくは **整合性** (consistency) の指標の1つである(両側検定の) **P値** (P-value) を複数定義して比較してみる。

以下ではそのP値を $\text{pvalue}(k|n, p = p_0)$ のように書くことにする。

データ「 n 回中当たりが k 回」が与えられたときに、パラメータ値 p に $\text{pvalue}(k|n, p = p_0)$ を対応させる関数

$$p \mapsto \text{pvalue}(k|n, p)$$

をデータ「 n 回中当たりが k 回」に関する **P値関数** (P-value function) と呼ぶ。

P値関数のグラフは、与えられたデータとパラメータ値 p の両立性の様子の全体の視覚化になっている。

$0 \leq \alpha \leq 1$ と仮定する。 α は **有意水準** (significance level) と呼ばれる。

データ「 n 回中当たりが k 回」のP値関数が与えられたとき、パラメータ p に関する **信頼度** (confidence level) $1 - \alpha$ の **信頼区間** (confidence interval) が次によって定義される：

$$\text{confint}(k|n, \alpha) = \{ p \mid \text{pvalue}(k|n, p) \geq \alpha \}.$$

すなわち、データ「 n 回中当たりが k 回」に関するパラメータ p の信頼度 $1 - \alpha$ の信頼区間は、データの数値とパラメータ値 p の両立性の指標であるP値が最初に決めておいた閾値 α 以上になるような値 $p = p_0$ 全体の集合である。

この集合が区間にならない場合には、この集合を含む最小の区間を信頼区間と定義することもある。

3 累積分布関数 cdf と分位点関数 quantile

確率分布 D に従う確率変数 X に対して、 $x \in \mathbb{R}$ の関数

$$\text{cdf}(D, x) = P(X \leq x)$$

を分布 D の **累積分布関数** (cumulative distribution function, **cdf**) と呼ぶ。

例えば、試行回数 n 、成功確率 p の二項分布 $\text{Binomial}(n, p)$ の累積分布関数は次のように表される：

$$\text{cdf}(\text{Binomial}(n, p), x) = \sum_{k \leq x} \binom{n}{k} p^k (1-p)^{n-k}.$$

例えば、平均 μ 、標準偏差 σ の正規分布 $\text{Normal}(\mu, \sigma)$ の累積分布関数は次のように表される：

$$\text{cdf}(\text{Normal}(\mu, \sigma), x) = \frac{1}{\sqrt{2\pi\sigma^2}} \int_{-\infty}^x \exp\left(-\frac{(t-\mu)^2}{2\sigma^2}\right) dt.$$

もしも、累積分布関数 $p = \text{cdf}(D, x)$ の逆関数が存在するならば、それを


$$x = \text{quantile}(D, p)$$

と表し、分布 D の **分位点関数** (quantile function) と呼ぶ。


例えば、正規分布の累積分布関数は狭義単調増加関数なので、分位点関数 $x = \text{quantile}(\text{Normal}(\mu, \sigma), p)$ が定義される。

累積分布関数の逆関数が存在しない場合にも、その代わりとなる便利な関数が定義可能な場合には、それも分位点関数と呼び、同じ記号で表す。


```
In [2]: 1 plot(x → cdf(Binomial(10, 0.3), x), -1.5, 11.5; label="")
2 plot!(xtick=-1:11, ytick=0:0.05:1)
3 plot!(xguide="x", yguide="p")
4 title!("p = cdf(Binomial(10, 0.3), x)")
```



```
In [3]: 1 plot(z → cdf(Normal(0,1), z), -4.5, 4.5; label="")
2 plot!(xtick=-10:10, ytick=0:0.1:1)
3 plot!(xguide="z", yguide="p")
4 title!("p = cdf(Normal(0,1), z)")
```



```
In [4]: 1 plot(p → quantile(Normal(0,1), p), 0, 1; label="")
2 plot!(ylim=(-4.5, 4.5))
3 plot!(ytick=-10:10, xtick=0:0.1:1)
4 plot!(yguide="z", xguide="p")
5 title!("z = quantile(Normal(0,1), p)")
```



3.1 補累積分布函数 ccdf と補分位点函数 cquantile

$\text{ccdf}(D, x) = 1 - \text{cdf}(D, x) = P(X > x)$ を **補累積分布関数** (complementary cumulative distribution function, **ccdf**) と呼ぶ.

補累積分布関数は単調減少関数になる.

補累積分布関数 $p = \text{ccdf}(D, x) = 1 - \text{cdf}(D, x)$ の逆関数を

$$x = \text{cquantile}(D, p) = \text{quantile}(D, 1 - p)$$

と表し, **補分位点関数** (complementary quantile function) と呼ぶ.

補分位点関数も単調減少関数になる.

例えば,

$$\text{cquantile}(\text{Normal}(0, 1), 0.05/2) = 1.95996398454 \dots \approx 1.96$$

は非常に有名である. 実践的にはこの値が 1.96 にぴったり等しいとして扱って問題ない.

In [5]: 1 quantile(Normal(0,1), 0.05/2)

Out[5]: 1.9599639845400592

4 WilsonのP値と信頼区間

WilsonのP値と信頼区間は二項分布の正規分布近似によって定義される.

4.1 WilsonのP値

k を二項分布 $\text{Binomial}(n, p)$ に従う確率変数とすると,

$$\hat{p} = \frac{k}{n}$$

とおくと, np と $n(1 - p)$ が十分に大きいならば, 中心極限定理より次の近似が成立している:

$$\frac{k - np}{\sqrt{np(1 - p)}} = \frac{\hat{p} - p}{\sqrt{p(1 - p)/n}} \sim \text{Normal}(0, 1), \text{ approximately.}$$

これを用いて, データ「 n 回中当たりは k 回」に関する「当たりが出る確率は p である」という仮説の **WilsonのP値** を次のように定義する:

$$\text{pvalue}_{\text{Wilson}}(k|n, p) = 2\text{ccdf}\left(\text{Normal}(0, 1), \left|\frac{\hat{p} - p}{\sqrt{p(1 - p)/n}}\right|\right).$$

4.2 Wilsonの信頼区間

WilsonのP値に対応する信頼度 $1 - \alpha$ の **Wilsonの信頼区間**

$$\text{confint}_{\text{Wilson}}(k|n, \alpha) = \{ p \mid \text{pvalue}_{\text{Wilson}}(k|n, p) \geq \alpha \}$$

がどうなるかを計算してみよう. これはP値関数から信頼区間がどのように得られるかを理解するために手頃な複雑さを持つ良い練習問題になっている.

$z_{\alpha/2}$ を次のように定義する:

$$z_{\alpha/2} = \text{cquantile}(\text{Normal}(0, 1), \alpha/2).$$

例えば, $\alpha = 5\%$ のとき, $z = z_{5\%/2} \approx 1.96$ になることは非常に有名である. このとき

$$\begin{aligned}
p \in \text{confint}_{\text{Wilson}}(k|n, \alpha) &\iff \text{pvalue}_{\text{Wilson}}(k|n, p) \geq \alpha \\
&\iff 2\text{ccdf}\left(\text{Normal}(0, 1), \left|\frac{\hat{p} - p}{\sqrt{p(1-p)/n}}\right|\right) \geq \alpha \\
&\iff \text{ccdf}\left(\text{Normal}(0, 1), \left|\frac{\hat{p} - p}{\sqrt{p(1-p)/n}}\right|\right) \geq \frac{\alpha}{2} \\
&\iff \left|\frac{\hat{p} - p}{\sqrt{p(1-p)/n}}\right| \leq \text{cquantile}\left(\text{Normal}(0, 1), \frac{\alpha}{2}\right) = z_{\alpha/2} \\
&\iff \frac{(p - \hat{p})^2}{p(1-p)/n} \leq z_{\alpha/2}^2 \\
&\iff (p - \hat{p})^2 \leq \frac{z_{\alpha/2}^2}{n} p(1-p) \\
&\iff \left(1 + \frac{z_{\alpha/2}^2}{n}\right)p^2 - 2\left(\hat{p} + \frac{z_{\alpha/2}^2}{2n}\right)p + \hat{p}^2 \leq 0.
\end{aligned}$$

ゆえに,

$$a = 1 + \frac{z_{\alpha/2}^2}{n}, \quad b = \hat{p} + \frac{z_{\alpha/2}^2}{2n}, \quad c = \hat{p}^2$$

とおき,

$$p_{\pm} = \frac{b \pm \sqrt{b^2 - ac}}{a}$$

とおくと, Wilsonの信頼区間は次のように表される:

$$\text{confint}_{\text{Wilson}}(k|n, \alpha) = [p_-, p_+].$$

注意: $f(p)$ を

$$f(p) = \left(1 + \frac{z_{\alpha/2}^2}{n}\right)p^2 - 2\left(\hat{p} + \frac{z_{\alpha/2}^2}{2n}\right)p + \hat{p}^2 = ap^2 - 2bp + c$$

とおくと,

$$f(0) = \hat{p}^2 \geq 0, \quad f(1) = (1 - \hat{p})^2 \geq 0$$

となるので, Wilsonの信頼区間は区間 $[0, 1]$ に自然に含まれることがわかる.

4.3 WilsonのP値と信頼区間の実装例

```

In [6]: 1 function pvalue_wilson(k, n, p)
        2    $\hat{p} = k/n$ 
        3    $se = \sqrt{p*(1 - p)/n}$ 
        4    $2\text{ccdf}(\text{Normal}(0,1), \text{abs}(\hat{p} - p)/se)$ 
        5 end
        6
        7 function confint_wilson(k, n,  $\alpha$ )
        8    $z = \text{cquantile}(\text{Normal}(0,1), \alpha/2)$ 
        9    $\hat{p} = k/n$ 
       10    $a, b, c = 1 + z^2/n, \hat{p} + z^2/(2n), \hat{p}^2$ 
       11    $\text{sqrtd} = \sqrt{b^2 - a*c}$ 
       12    $(b - \text{sqrtd})/a, (b + \text{sqrtd})/a$ 
       13 end

```

Out[6]: confint_wilson (generic function with 1 method)


In [7]:

```
1 @show n, k, α = 10, 3, 0.05
2 @show ci = confint_wilson(k, n, α) ▷ collect
3
4 P_Wilson = plot(p → pvalue_wilson(k, n, p), 0, 1; label="")
5 plot!(xguide="p", yguide="P-value")
6 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
7 plot!(ci, fill(α,2); label="$ (100(1-α))% CI")
8 title!("Wilson's P-value function for n=$n, k=$k")
```

(n, k, α) = (10, 3, 0.05) = (10, 3, 0.05)

ci = confint_wilson(k, n, α) ▷ collect = [0.10779126740630075, 0.6032218525388553]

Out[7]:



このように信頼度 $1 - \alpha$ の信頼区間はP値函数のグラフを高さ α で切断して得られる区間になる。

上のグラフは次のように解釈される。

- データ「 $n = 10$ 回中当たりが $k = 3$ 回」と最も **相性がよい** パラメータ値は $p = 0.3$ である(点推定)。
- 0 近くの p や 0.8 以上の p は「 $n = 10$ 回中当たりが $k = 3$ 回」というデータと **両立しない**。
- 閾値を $\alpha = 5\%$ に定めておいて、P値が α 未満になったときに「パラメータ値とデータの数値が両立しなさすぎる」と判断することにしたとき、そのように判断されずにすむ p の範囲はおおよそ 10.8% から 60.3% になる(信頼区間)。

これが 95% 信頼区間の穏健な(自信過剰にならずにすむ)解釈になる。

5 WaldのP値と信頼区間

WaldのP値と信頼区間は、WilsonのP値と信頼区間の構成における二項分布における $\hat{p} = k/n$ の分散 $p(1 - p)/n$ をその推定値 $\hat{p}(1 - \hat{p})/n$ で置き換えることによって定義される。

これによって、WaldのP値と信頼区間の誤差はWilsonのP値と信頼区間よりも大きくなってしまう。

その代わりに、Waldの信頼区間はWilsonの信頼区間と違って二次方程式を解くことなく容易に計算可能になる。

5.1 WaldのP値

データ「 n 回中当たりは k 回」に関する「当たりが出る確率は p である」という仮説の **WaldのP値** を次のように定義する:

$$\text{pvalue}_{\text{Wald}}(k|n, p) = 2\text{ccdf}\left(\text{Normal}(0, 1), \left| \frac{\hat{p} - p}{\sqrt{\hat{p}(1 - \hat{p})/n}} \right| \right).$$

5.2 Waldの信頼区間

WaldのP値に対応する信頼度 $1 - \alpha$ の **Waldの信頼区間**

$$\text{confint}_{\text{Wald}}(k|n, \alpha) = \{ p \mid \text{pvalue}_{\text{Wald}}(k|n, p) \geq \alpha \}$$

がどうなるかを計算してみよう. 以下のようにWilsonの信頼区間の場合よりもずっと簡単になる.

$$\begin{aligned} p \in \text{confint}_{\text{Wald}}(k|n, \alpha) &\iff \text{pvalue}_{\text{Wald}}(k|n, p) \geq \alpha \\ &\iff 2\text{ccdf}\left(\text{Normal}(0, 1), \left| \frac{\hat{p} - p}{\sqrt{\hat{p}(1 - \hat{p})/n}} \right| \right) \geq \alpha \\ &\iff \text{ccdf}\left(\text{Normal}(0, 1), \left| \frac{\hat{p} - p}{\sqrt{\hat{p}(1 - \hat{p})/n}} \right| \right) \geq \frac{\alpha}{2} \\ &\iff \left| \frac{\hat{p} - p}{\sqrt{\hat{p}(1 - \hat{p})/n}} \right| \leq \text{cquantile}\left(\text{Normal}(0, 1), \frac{\alpha}{2}\right) = z_{\alpha/2} \\ &\iff \hat{p} - z_{\alpha/2} \sqrt{\frac{p(1 - p)}{n}} \leq p \leq \hat{p} + z_{\alpha/2} \sqrt{\frac{p(1 - p)}{n}}. \end{aligned}$$

すなわち,

$$\text{confint}_{\text{Wald}}(k|n, \alpha) = \left[\hat{p} - z_{\alpha/2} \sqrt{\frac{p(1 - p)}{n}}, \hat{p} + z_{\alpha/2} \sqrt{\frac{p(1 - p)}{n}} \right].$$

Waldの信頼区間は区間 $[0, 1]$ をはみ出すこともある.

5.3 WaldのP値と信頼区間の実装例

```
In [8]: 1 function pvalue_wald(k, n, p)
2       p_hat = k/n
3       sehat = sqrt(p_hat*(1 - p_hat)/n)
4       2ccdf(Normal(0,1), abs(p_hat - p)/sehat)
5   end
6
7   function confint_wald(k, n, alpha)
8       z = cquantile(Normal(0,1), alpha/2)
9       p_hat = k/n
10      sehat = sqrt(p_hat*(1 - p_hat)/n)
11      p_hat - z*sehat, p_hat + z*sehat
12  end
```


Out[8]: confint_wald (generic function with 1 method)

```
In [9]: 1 @show n, k, alpha = 10, 3, 0.05
2 @show ci = confint_wald(k, n, alpha) > collect
3
4 P_Wald = plot(p -> pvalue_wald(k, n, p), -0.1, 1.1; label="")
5 plot!(xguide="p", yguide="P-value")
6 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
7 plot!(ci, fill(alpha, 2); label="$ (100(1-alpha))% CI")
8 title!("Wald's P-value function for n=$n, k=$k")
```

(n, k, alpha) = (10, 3, 0.05) = (10, 3, 0.05)

ci = confint_wald(k, n, alpha) > collect = [0.0159742349106739, 0.5840257650893261]

Out[9]: Wald's P-value function for n=10, k=3



WaldのP値関数は左右対称になり, 一般に台(0 でない値を取る範囲)が区間 $[0, 1]$ をはみ出す.

WaldのP値関数とそれに付随する信頼区間はこうのように誤差が大きくなってしまふものだが、信頼区間の計算が簡単になるという利点がある。

np と $n(1-p)$ が非常に大きければ、Waldの信頼区間はWilsonの信頼区間とほぼぴったり一致するようになり、二次方程式を解いてWilsonの信頼区間を求めるメリットはなくなる。

コンピュータで自動的に計算するならば、Wilsonの信頼区間の方を計算するべきだが、暗算でどんぶり勘定したい場合にはWaldの信頼区間の方が便利である。

このように統計学的道具にはそれぞれ異なる利点と欠点があり、適材適所で使い分けるのがよい。

6 Clopper-PearsonのP値と信頼区間

二項分布における確率を正規分布で近似せず、正確に求めることによって定義されるP値と信頼区間の一種がClopper-PearsonのP値と信頼区間である。

6.1 Clopper-PearsonのP値

データ「 n 回中当たりは k 回」に関する「当たりが出る確率は p である」という仮説の **Clopper-PearsonのP値** を次のように定義する：

$$pvalue_{CP}(k|n, p) = \min \left(\begin{array}{c} 1 \\ 2cdf(\text{Binomial}(n, p), k) \\ 2ccdf(\text{Binomial}(n, p), k-1) \end{array} \right).$$

ここで、 $cdf(\text{Binomial}(n, p), k)$ と $ccdf(\text{Binomial}(n, p), k-1)$ はそれぞれ、二項分布において k 以下になる確率と k 以下になる確率なので、この定義はおおよそ「片側検定のP値の2倍」になっている。その定義のままだと1より大きくなる場合があるので、1以下になるようにしてある。

6.2 二項分布とベータ分布の関係

Clopper-PearsonのP値に対応する信頼度 $1 - \alpha$ のClopper-Pearsonの信頼区間を計算するためには、二項分布とベータ分布の素晴らしい関係が必要になる。次が成立している：

$$\begin{aligned} cdf(\text{Binomial}(n, p), k) &= ccdf(\text{Beta}(k+1, n-k), p) \quad (0 \leq k < n), \\ ccdf(\text{Binomial}(n, p), k-1) &= cdf(\text{Beta}(k, n-k+1), p) \quad (0 < k \leq n). \end{aligned}$$

前者は後者で k を $k+1$ に置き換えれば得られるので、後者のみを示そう。

後者は具体的には次が成立するということを意味している：

$$\sum_{i \geq k} \binom{n}{i} p^i (1-p)^{n-i} = \frac{1}{B(k, n-k+1)} \int_0^p t^{k-1} (1-t)^{n-k} dt.$$

両辺の $p=0$ での値は0になる。

ゆえに両辺の p に関する導関数が等しいことを示せば、両辺が等しいことがわかる。

左辺を p で微分すると以下ようになる：

$$\begin{aligned} \frac{\partial}{\partial p} \sum_{i \geq k} \binom{n}{i} p^i (1-p)^{n-i} &= \sum_{i \geq k} \binom{n}{i} i p^{i-1} (1-p)^{n-i} - \sum_{i \geq k} \binom{n}{i} (n-i) p^{i-1} (1-p)^{n-i-1} \\ &= \sum_{i \geq k} \frac{n!}{(i-1)!(n-i)!} p^{i-1} (1-p)^{n-i} - \sum_{i \geq k} \frac{n!}{i!(n-i-1)!} p^{i-1} (1-p)^{n-i-1} \\ &= \sum_{i \geq k} \frac{n!}{(i-1)!(n-i)!} p^{i-1} (1-p)^{n-i} - \sum_{i \geq k+1} \frac{n!}{(i-1)!(n-i)!} p^{i-1} (1-p)^{n-i-1} \\ &= \frac{n!}{(k-1)!(n-k)!} p^{k-1} (1-p)^{n-k} \end{aligned}$$


右辺を p で微分すると

$$\begin{aligned}
\frac{1}{B(k, n-k+1)} \frac{\partial}{\partial p} \int_0^p t^{k-1} (1-t)^{n-k} dt &= \frac{1}{B(k, n-k+1)} p^{k-1} (1-p)^{n-k} \\
&= \frac{\Gamma(k + (n-k+1))}{\Gamma(k) \Gamma(n-k+1)} p^{k-1} (1-p)^{n-k} \\
&= \frac{n!}{(k-1)! (n-k)!} p^{k-1} (1-p)^{n-k}.
\end{aligned}$$


これで左辺と右辺の導関数が等しいことが証明された。

注意: 二項分布の累積分布関数と補累積分布関数は素朴に和で実装するよりも、上の公式を用いて、ベータ分布の補累積分布関数と累積分布関数で計算するようにした方が効率が良い。ベータ分布の累積分布関数は *regularized incomplete Beta function* としてコンピュータの基本特殊関数ライブラリで効率的に実装されている。

```
In [10]: 1 n, k = 10, 3
2 plot(p → ccdf(Binomial(n,p), k-1);
3       label="ccdf(Binomial(n,p), k-1)")
4 plot!(p → cdf(Beta(k, n-k+1), p), 0, 1;
5       label="cdf(Beta(k, n-k+1), p)", ls=:dash)
6 plot!(legend=:bottomright)
7 plot!(xguide="p")
8 plot!(xtick=0:0.1:1, ytick=0:0.1:1)
9 title!("n=$n, k=$k")
```



```
In [11]: 1 n, k = 10, 3
2 plot(p → cdf(Binomial(n,p), k);
3       label="cdf(Binomial(n,p), k)")
4 plot!(p → ccdf(Beta(k+1, n-k), p), 0, 1;
5       label="ccdf(Beta(k+1, n-k), p)", ls=:dash)
6 plot!(legend=:topright)
7 plot!(xguide="p")
8 plot!(xtick=0:0.1:1, ytick=0:0.1:1)
9 title!("n=$n, k=$k")
```



6.3 Clopper-Pearsonの信頼区間

Clopper-PearsonのP値に対応する信頼度 $1 - \alpha$ の Clopper-Pearsonの信頼区間

$$\text{confint}_{\text{CP}}(k|n, \alpha) = \{ p \mid \text{pvalue}_{\text{CP}}(k|n, p) \geq \alpha \}$$

がどうなるかを計算してみよう。

$$\begin{aligned} p &\in \text{confint}_{\text{CP}}(k|n, \alpha) \\ &\iff \text{pvalue}_{\text{CP}}(k|n, p) \geq \alpha \\ &\iff \text{cdf}(\text{Binomial}(n, p), k) \geq \alpha/2 \text{ and } \text{ccdf}(\text{Binomial}(n, p), k-1) \geq \alpha/2 \\ &\iff \text{ccdf}(\text{Beta}(k+1, n-k), p) \geq \alpha/2 \text{ and } \text{cdf}(\text{Beta}(k, n-k+1), p) \geq \alpha/2 \\ &\iff p \leq \text{cquantile}(\text{Beta}(k+1, n-k), \alpha/2) \text{ and } p \geq \text{quantile}(\text{Beta}(k, n-k+1), \alpha/2), \end{aligned}$$

ゆえに

$$p_L = \text{quantile}(\text{Beta}(k, n-k+1), \alpha/2), \quad p_U = \text{cquantile}(\text{Beta}(k+1, n-k), \alpha/2)$$

とおくと,

$$\text{confint}_{\text{CP}}(k|n, \alpha) = [p_L, p_U].$$

6.4 Clopper-PearsonのP値と信頼区間の実装例

```
In [12]: 1 function pvalue_cp(k, n, p)
2         bin = Binomial(n, p)
3         min(1, 2cdf(bin, k), 2ccdf(bin, k-1))
4     end
5
6     function confint_cp(k, n, α)
7         quantile(Beta(k, n-k+1), α/2), cquantile(Beta(k+1, n-k), α/2)
8     end
```


Out[12]: confint_cp (generic function with 1 method)

```
In [13]: 1 @show n, k, α = 10, 3, 0.05
2 @show ci = confint_cp(k, n, α) ▷ collect
3
4 P_CP = plot(p → pvalue_cp(k, n, p), 0, 1; label="")
5 plot!(xguide="p", yguide="P-value")
6 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
7 plot!(ci, fill(α, 2); label="$ (100(1-α))% CI")
8 title!("Clopper-Pearson P-value function for n=$n, k=$k")
```

(n, k, α) = (10, 3, 0.05) = (10, 3, 0.05)

ci = confint_cp(k, n, α) ▷ collect = [0.06673951117773438, 0.6524528500599971]

Out[13]: Clopper-Pearson P-value function for n=10, k=3



Clopper-Pearsonの信頼区間は小さめの n については過剰に保守的になることが多い。

しかし、上のコードを見れば分かるように、実装は極めてシンプルである。

そして、さらにデータが二項分布 $\text{Binomial}(n, p)$ に従ってランダムに生成されているとき、P値が α 以下になる確率が確実に α 以下になってくれるという利点もある。

7 SterneのP値と信頼区間

SterneのP値と信頼区間も、二項分布における確率を正規分布で近似せずに正確に求めることによって定義される。Sterneの信頼区間の方がClopper-Pearsonの信頼区間よりも概ね狭くなってくるといえる利点があるが、Clopper-Pearsonの信頼区間の場合と違ってシンプルな公式がないので、実装はかなり面倒になる。

7.1 SterneのP値と信頼区間

データ「 n 回中あたりは k 回」に関する「当たりが出る確率は p である」という仮説の **SterneのP値** を次のように定義する:

$$\text{pvalue}_{\text{Sterne}}(k|n, p) = \sum_{P(i|n, p) \leq P(k|n, p)} P(i|n, p).$$

ここで,

$$P(i|n, p) = \binom{n}{i} p^i (1-p)^{n-i} \quad (i = 0, 1, \dots, n).$$

SterneのP値は二項分布 $\text{Binomial}(n, p)$ 内でデータの数値よりも生じる確率が低い値の確率の総和になる。

これに対応するSterneの信頼区間は

$$\text{confint}_{\text{Sterne}}(k|n, \alpha) = \{ p \mid \text{pvalue}_{\text{Sterne}}(k|n, p) \geq \alpha \}$$

であるが、この定義のままだと区間にならない場合があるので、実装時にはこの集合を含む最小の区間として実装することにする。

7.2 SterneのP値と信頼区間の実装例

効率的でない実装でよいならば、P値関数の定義は、

```
In [14]: 1 x ≲ y = x < y || x ≈ y
2
3 function pvalue_sterne_naive(k, n, p)
4     bin = Binomial(n, p)
5     sum(pdf(bin, i) for i in support(bin) if pdf(bin, i) ≲ pdf(bin, k))
6 end
```

```
Out[14]: pvalue_sterne_naive (generic function with 1 method)
```

で十分だが、次のセルのように計算は非常に遅くなる。

```
In [15]: 1 @time pvalue_sterne_naive(3*10^6, 10*10^6, 0.301)
2 @time pvalue_sterne_naive(3*10^6, 10*10^6, 0.301)
3 @time pvalue_sterne_naive(3*10^6, 10*10^6, 0.301)

2.982381 seconds
3.007138 seconds
3.026064 seconds
```

```
Out[15]: 5.354898851685295e-12
```

```

In [16]: 1 x ≈ y = x < y || x ≈ y
2
3 _pdf_le(x, (dist, y)) = pdf(dist, x) ≈ y
4
5 function _search_boundary(f, x0, Δx, param)
6     x = x0
7     if f(x, param)
8         while f(x - Δx, param) x -= Δx end
9     else
10        x += Δx
11        while !f(x, param) x += Δx end
12    end
13    x
14 end
15
16 function pvalue_sterne(dist::DiscreteUnivariateDistribution, x)
17     Px = pdf(dist, x)
18     Px == 0 && return Px
19     Px == 1 && return Px
20     m = mode(dist)
21     Px ≈ pdf(dist, m) && return one(Px)
22     if x < m
23         y = _search_boundary(_pdf_le, 2m - x, 1, (dist, Px))
24         cdf(dist, x) + ccdf(dist, y-1)
25     else # x > m
26         y = _search_boundary(_pdf_le, 2m - x, -1, (dist, Px))
27         cdf(dist, y) + ccdf(dist, x-1)
28     end
29 end
30
31 function pvalue_sterne(k, n, p)
32     pvalue_sterne(Binomial(n, p), k)
33 end
34
35 function confint_sterne(k, n, α)
36     a, b = confint_cp(k, n, α/10)
37     ps = find_zeros(a-√eps(), b+√eps()) do p
38         logit(0 < p ≤ 1 ? pvalue_sterne(k, n, p) : zero(p)) - logit(α)
39     end
40     first(ps), last(ps)
41 end

```

Out[16]: confint_sterne (generic function with 1 method)

```

In [17]: 1 @time pvalue_sterne(3*10^6, 10*10^6, 0.301)
2 @time pvalue_sterne(3*10^6, 10*10^6, 0.301)
3 @time pvalue_sterne(3*10^6, 10*10^6, 0.301)

```

```

0.000049 seconds (8 allocations: 1.875 KiB)
0.000010 seconds (8 allocations: 1.875 KiB)
0.000013 seconds (8 allocations: 1.875 KiB)

```

Out[17]: 5.354898860502031e-12

```

In [18]: 1 @time confint_sterne(3*10^6, 10*10^6, 0.001)
2 @time confint_sterne(3*10^6, 10*10^6, 0.001)
3 @time confint_sterne(3*10^6, 10*10^6, 0.001)

```

```

0.142895 seconds (51.68 k allocations: 3.764 MiB, 98.03% compilation time)
0.002759 seconds (11.48 k allocations: 2.402 MiB)
0.003139 seconds (11.48 k allocations: 2.402 MiB)

```

Out[18]: (0.2995232976841354, 0.30047704792862706)

```

In [19]: 1 @show n, k,  $\alpha$  = 10, 3, 0.05
          2 @show ci = confint_sterne(k, n,  $\alpha$ ) > collect
          3
          4 P_Sterne = plot(p → pvalue_sterne(k, n, p), 0, 1; label="")
          5 plot!(xguide="p", yguide="P-value")
          6 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
          7 plot!(ci, fill( $\alpha$ , 2); label="$ (100(1- $\alpha$ ))% CI")
          8 title!("Sterne's P-value function for n=$n, k=$k")


```

```

(n, k,  $\alpha$ ) = (10, 3, 0.05) = (10, 3, 0.05)
ci = confint_sterne(k, n,  $\alpha$ ) > collect = [0.0872644339141502, 0.6194106589132863]


```

Out[19]:



8 4種のP値関数と信頼区間を同時にプロット

Out[20]:



9 ベイズ統計での信用区間に対応するP値関数

P値関数と区間推定の表裏一体性については、私のツイログの検索を参照せよ:

- P値 信頼区間 表裏一体 (<https://twilog.org/genkuroki/search?word=P%E5%80%A4%20%E4%BF%A1%E9%A0%BC%E5%8C%BA%E9%96%93%20%E8%A1%A8%E8%A3%8F%E4%B8%>)

区間推定の方法には、通常の信頼区間以外にも、ベイズ統計の **信用区間** (credible interval, 確信区間)の方法もある。

ゆえに、P値関数と区間推定の表裏一体性より、ベイズ統計の方法を使っても、P値関数を構成することができる。

ベイズ版のP値関数と通常のP値関数を比較したい。

```
1 ### 二項分布モデルのHDI版信用区間
2
3 簡単のため共役事前分布の場合のみを扱う。
4
5 二項分布モデルのベイズ統計では以下の手続きでパラメータ  $p$  の事後分布が得られる。
6
7 (1) パラメータ  $p$  の事前分布  $\text{prior} = \text{Beta}(a, b)$  ( $a, b > 0$ ) を任意に用意する。
8
9 (2) 「 $n$  回中当たりが  $k$  回」の型のデータを取得する。
10
11 (3) パラメータ  $p$  の事後分布  $\text{posterior} = \text{Beta}(k+a, n-k+b)$  が得られる。
12
13 この手続きの中身については任意のベイズ統計に関する教科書を参照せよ。
14
15 このとき、パラメータ  $p$  に関する  $100(1-\alpha)\%$  信用区間が次のように定義される。
16
17 (4) 事後分布を
18
19  $[0, 1]$  に含まれる区間  $[p_L, p_U]$  で事後分布で測ったその区間に含まれる確率が
20
21
22 
$$\text{cdf}(\text{posterior}, p_U) - \text{cdf}(\text{posterior}, p_L) = 1 - \alpha$$

23
24
25 となるものの中で長さ  $p_U - p_L$  が最小のものを highest density interval (HDI) として定義され
26 た信用度  $1-\alpha$  の __信用区間__ と呼び、次のように表す：
27
28 
$$\text{credint\_HDI}(k|n, p, a, b) = [p_L, p_U].$$

29
30
31 ベータ分布の確率密度関数を
32
33
34 
$$\text{pdf}(\text{Beta}(a, b), t) = \frac{1}{B(a, b)} t^{a-1} (1-t)^{b-1}$$

35
36
37 と書くと、HDI版信用区間  $[p_L, p_U]$  について、
38
39
40 
$$\text{pdf}(\text{Beta}(k+a, n-k+b), p_L) =$$

41 
$$\text{pdf}(\text{Beta}(k+a, n-k+b), p_U)$$

42
43
44 が成立しており、この条件と上の確率に関する条件でHDI版信用区間は一意に特徴付けられる。
45
46 __注意:__ 実際の実装では、
47
48
49  $p_L = \text{quantile}(\text{posterior}, t), \quad \text{quad}$ 
50  $p_U = \text{quantile}(\text{posterior}, t + (1 - \alpha))$ 
51
52
53 のあいだの長さ、
54
55
56  $f(t) = p_U - p_L =$ 
57  $\text{quantile}(\text{posterior}, t + (1 - \alpha)) - \text{quantile}(\text{posterior}, t)$ 
58
59
60 を最小化する  $t$  を求め、それに対応する区間  $[p_L, p_U]$  としてHDI版信用区間を計算する。
```

```
1 ### HDI版信用区間に対応するP値関数
2
3 HDI版信用区間に対応するP値関数  $\text{pvalue\_HDI}(k|n, p, a, b)$  は以下のように定義される。
4
5 与えられた値  $p$  に対して、 $t$  に関する方程式
6
7
8 
$$\text{pdf}(\text{Beta}(k+a, n-k+b), t) =$$

9 
$$\text{pdf}(\text{Beta}(k+a, n-k+b), p)$$

10
```

```

10 ~~~~~
11
12 の2つの解を  $p_L$  と  $p_U$  と書き,
13
14 $$
15 \text{pvalue\_HDI}(k|n,p,a,b) =
16 \text{cdf}(\text{Beta}(k+a, n-k+b), p_L) +
17 \text{ccdf}(\text{Beta}(k+a, n-k+b), p_U).
18 $$
19
20 これを, __ベイズ版P値函数__ (Bayesian P-value function)のHDI版と呼ぶ.

```

1 ### HDI版の信用区間とそれに対応するベイズ版P値函数の実装例

```

In [27]: 1 function hdi(dist::ContinuousUnivariateDistribution, α = 0.05; alg = Brent())
2         f(p) = quantile(dist, p + (1 - α)) - quantile(dist, p)
3         o = optimize(f, 0, α, alg)
4         p = o.minimizer
5         quantile.(dist, (p, p + (1 - α)))
6     end
7
8     function pvalue_hdi(dist::ContinuousUnivariateDistribution, x₀; xlim = extrema(dist))
9         p₀ = pdf(dist, x₀)
10        m = mode(dist)
11        f(x) = pdf(dist, x) - p₀
12        if x₀ == m
13            1.0
14        elseif x₀ > m
15            x₁ = find_zero(f, (xlim[begin], m))
16            cdf(dist, x₁) + ccdf(dist, x₀)
17        else
18            x₁ = find_zero(f, (m, xlim[end]))
19            cdf(dist, x₀) + ccdf(dist, x₁)
20        end
21    end
22
23    function pvalue_hdi(k, n, p; a=1, b=1)
24        posterior = Beta(k+a, n-k+b)
25        pvalue_hdi(posterior, p)
26    end
27
28    function credint_hdi(k, n, α; a=1, b=1)
29        posterior = Beta(k+a, n-k+b)
30        hdi(posterior, α)
31    end

```

Out[27]: credint_hdi (generic function with 1 method)


```

In [30]: 1 @show n, k,  $\alpha$  = 10, 3, 0.05
2 @show a, b = 1, 1
3 @show ci = credint_hdi(k, n,  $\alpha$ ; a, b) ▷ collect
4
5 P_HDI = plot(p → pvalue_hdi(k, n, p; a, b), 0, 1; label="")
6 plot!(xguide="p", yguide="P-value")
7 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
8 plot!(ci, fill( $\alpha$ , 2); label="$ (100(1- $\alpha$ ))% CI (HDI)")
9 title!("Bayesian P-value function for n=$n, k=$k, a=$a, b=$b")


```

(n, k, α) = (10, 3, 0.05) = (10, 3, 0.05)

(a, b) = (1, 1) = (1, 1)

ci = credint_hdi(k, n, α ; a, b) ▷ collect = [0.09337233320249291, 0.5879525593275807]

Out[30]:



このグラフも通常のP値関数と信頼区間と同様に、

- データの数値とパラメータ値の相性の良さ(両立性, compatibility)

を表すグラフだと解釈可能である. この解釈は次の事後分布のグラフにもそのまま一般化される.

```

In [31]: 1 @show n, k,  $\alpha$  = 10, 3, 0.05
2 @show a, b = 1, 1
3 @show ci = credint_hdi(k, n,  $\alpha$ ; a, b) ▷ collect
4
5 f(p) = pdf(Beta(k+a, n-k+b), p)
6 Q_HDI = plot(f, 0, 1; label="")
7 plot!(xguide="p", yguide="probability density")
8 plot!(xtick=0:0.1:1)
9 plot!(ci, fill(f(ci[end]), 2); label="$ (100(1- $\alpha$ ))% CI (HDI)")
10 title!("Posterior for n=$n, k=$k, a=$a, b=$b")


```

(n, k, α) = (10, 3, 0.05) = (10, 3, 0.05)

(a, b) = (1, 1) = (1, 1)

ci = credint_hdi(k, n, α ; a, b) ▷ collect = [0.09337233320249291, 0.5879525593275807]

Out[31]:



9.4 WilsonのP値関数とベイズ版P値関数のHDI版の比較

この節の目的は通常のP値関数とベイズ版P値関数を比較することであった.


以下では、通常のP値関数の代表例としてWilsonのP値関数を選び、ベイズ版P値関数の代表例として一様事前分布 $\text{Beta}(1, 1)$ から得られるベイズ版P値関数を選んで比較することにする。

上のベイズ版P値関数をWilsonのP値関数と重ねてプロットしてみよう。

```
In [32]: 1 @show n, k,  $\alpha$  = 10, 3, 0.05
2 @show a, b = 1, 1
3
4 plot(p  $\rightarrow$  pvalue_hdi(k, n, p; a, b), 0, 1; label="Bayesian (HDI)")
5 plot!(p  $\rightarrow$  pvalue_wilson(k, n, p); label="Wilson", ls=:dash)
6 plot!(xguide="p", yguide="P-value")
7 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
8 title!("P-value functions for n=$n, k=$k, a=$a, b=$b")
```

$(n, k, \alpha) = (10, 3, 0.05) = (10, 3, 0.05)$
 $(a, b) = (1, 1) = (1, 1)$

Out[32]: P-value functions for n=10, k=3, a=1, b=1




ベイズ版P値関数が通常のWilsonのP値関数によく一致していることが確認された。

n, k を大きくすると違いはさらに小さくなる。

```
In [33]: 1 @show n, k,  $\alpha$  = 30, 9, 0.05
2 @show a, b = 1, 1
3
4 plot(p  $\rightarrow$  pvalue_hdi(k, n, p; a, b), 0.05, 0.6; label="Bayesian (HDI)")
5 plot!(p  $\rightarrow$  pvalue_wilson(k, n, p); label="Wilson", ls=:dash)
6 plot!(xguide="p", yguide="P-value")
7 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
8 title!("P-value functions for n=$n, k=$k, a=$a, b=$b")
```

$(n, k, \alpha) = (30, 9, 0.05) = (30, 9, 0.05)$
 $(a, b) = (1, 1) = (1, 1)$


Out[33]: P-value functions for n=30, k=9, a=1, b=1



```
In [34]: 1 @show n, k, α = 100, 30, 0.05
2 @show a, b = 1, 1
3
4 plot(p → pvalue_hdi(k, n, p; a, b), 0.15, 0.5; label="Bayesian (HDI)")
5 plot!(p → pvalue_wilson(k, n, p); label="Wilson", ls=:dash)
6 plot!(xguide="p", yguide="P-value")
7 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
8 title!("P-value functions for n=$n, k=$k, a=$a, b=$b")
```

```
(n, k, α) = (100, 30, 0.05) = (100, 30, 0.05)
(a, b) = (1, 1) = (1, 1)
```

Out[34]: P-value functions for n=100, k=30, a=1, b=1



$n = 100, k = 30$ ではほぼびったり重なっている!

9.5 ETI版の信用区間とそれに対応するP値函数

HDI版ではなく, equal tailed interval (ETI)版の信用区間も定義しておこう:

$$\text{credint}_{\text{ETI}}(k|n, p, a, b) = [\text{quantile}(\text{posterior}, \alpha/2), \text{cquantile}(\text{posterior}, \alpha/2),]$$

ここで, $\text{posterior} = \text{Beta}(k + a, n - k + b)$.

これに対応するP値函数は次になる:

$$\text{pvalue}_{\text{ETI}}(k|n, p, a, b) = \min \begin{pmatrix} 1 \\ 2\text{cdf}(\text{posterior}, p) \\ 2\text{ccdf}(\text{posterior}, p) \end{pmatrix}.$$

9.6 ETI版の信用区間とそれに対応するベイズ版P値函数の実装例

```
In [29]: 1 function pvalue_eti(k, n, p; a=1/2, b=1/2)
2     posterior = Beta(k+a, n-k+a)
3     min(1, 2cdf(posterior, p), 2ccdf(posterior, p))
4 end
5
6 function credint_eti(k, n, α; a=1/2, b=1/2)
7     posterior = Beta(k+a, n-k+a)
8     quantile(posterior, α/2), cquantile(posterior, α/2)
9 end
```

Out[29]: credint_eti (generic function with 1 method)


```
In [38]: 1 @show n, k, α = 10, 3, 0.05
2 @show a, b = 1/2, 1/2
3 @show ci = credint_eti(k, n, α; a, b) ▷ collect
4
5 P_ETI = plot(p → pvalue_eti(k, n, p; a, b), 0, 1; label="")
6 plot!(xguide="p", yguide="P-value")
7 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
8 plot!(ci, fill(α,2); label="$ (100(1-α))% CI (ETI)")
9 title!("Bayesian P-value function for n=$n, k=$k, a=$a, b=$b")
```

(n, k, α) = (10, 3, 0.05) = (10, 3, 0.05)

(a, b) = (1 / 2, 1 / 2) = (0.5, 0.5)

ci = credint_eti(k, n, α; a, b) ▷ collect = [0.09269459393815314, 0.6058183181486713]

Out[38]: Bayesian P-value function for n=10, k=3, a=0.5, b=0.5



9.7 WilsonのP値関数とベイズ版P値関数のETI版の比較

ベイズ版P値関数のETI版は, Jeffreys事前分布 Beta(1/2, 1/2) の場合にWilsonのP値とよく一致することを確認しよう。


Jeffreys事前分布は一様事前分布と共に **無情報事前分布** の代表例である。

```
In [37]: 1 @show n, k, α = 10, 3, 0.05
2 @show a, b = 1/2, 1/2
3
4 plot(p → pvalue_eti(k, n, p; a, b), 0, 1; label="Bayesian (ETI)")
5 plot!(p → pvalue_wilson(k, n, p); label="Wilson", ls=:dash)
6 plot!(xguide="p", yguide="P-value")
7 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
8 title!("P-value functions for n=$n, k=$k, a=$a, b=$b")
```

(n, k, α) = (10, 3, 0.05) = (10, 3, 0.05)

(a, b) = (1 / 2, 1 / 2) = (0.5, 0.5)


Out[37]: P-value functions for n=10, k=3, a=0.5, b=0.5



```
In [40]: 1 @show n, k,  $\alpha$  = 30, 9, 0.05
2 @show a, b = 1/2, 1/2
3
4 plot(p  $\rightarrow$  pvalue_eti(k, n, p; a, b), 0.05, 0.6; label="Bayesian (ETI)")
5 plot!(p  $\rightarrow$  pvalue_wilson(k, n, p); label="Wilson", ls=:dash)
6 plot!(xguide="p", yguide="P-value")
7 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
8 title!("P-value functions for n=$n, k=$k, a=$a, b=$b")
```

(n, k, α) = (30, 9, 0.05) = (30, 9, 0.05)
(a, b) = (1 / 2, 1 / 2) = (0.5, 0.5)


Out[40]: P-value functions for n=30, k=9, a=0.5, b=0.5



```
In [42]: 1 @show n, k,  $\alpha$  = 100, 30, 0.05
2 @show a, b = 1/2, 1/2
3
4 plot(p  $\rightarrow$  pvalue_eti(k, n, p; a, b), 0.15, 0.5; label="Bayesian (ETI)")
5 plot!(p  $\rightarrow$  pvalue_wilson(k, n, p); label="Wilson", ls=:dash)
6 plot!(xguide="p", yguide="P-value")
7 plot!(xtick=0:0.1:1, ytick=0:0.05:1)
8 title!("P-value functions for n=$n, k=$k, a=$a, b=$b")
```

(n, k, α) = (100, 30, 0.05) = (100, 30, 0.05)
(a, b) = (1 / 2, 1 / 2) = (0.5, 0.5)

Out[42]: P-value functions for n=100, k=30, a=0.5, b=0.5



$n = 100, k = 30$ ではほぼぴったり重なっている!

9.8 道具を主義の色眼鏡で見ることの不合理性

所謂「ベイズ主義」にこだわったベイズ統計の解説では、通常の信頼区間とベイズ版の信用区間は解釈が異なり、まったく違うものであるかのように説明されている。

しかし、上で確認したように、二項分布モデルのような統計学入門の教科書でよく説明されているシンプルモデルの場合には、ベイズ版の信用区間は通常の信頼区間と実用的には区別する必要がないほどよく一致する。

そのことは、対応するP値関数がよく一致することと数学的には同値である。

むしろ、通常の4種のP値関数のあいだの違いの方が、ベイズ版P値関数とWilsonのP値関数の違いよりも大きいと言える。

この事実は、ベイズ版信用区間が通常の信頼区間の近似計算法だとみなせ、逆に、通常の信頼区間は無情報事前分布のベイズ版信用区間の近似計算法だとみなせることを意味している。

実践的には「同じもの」とみなしても害がないのに、主義の違いによって異なる解釈を強制しようとするのは、「かけ算の意味によって、 3×4 は正解だが、 4×3 は誤りになる」と教えている困った人達と同じ立場に立つことになる。

そういうことはもうやめた方がよいだろう。

「シンプルなモデルにおいては無情報事前分布のベイズ統計と通常のP値を使う方法は実践的には同等だとみなされる」という事実を理解して、使える道具の幅を気軽に広げた方が生産的と思われる。

それを妨げているのは「主義の押し付け」であり、極めて有害と思われる。

In []:

1	
---	--